Rules for integrands of the form $(d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p$

1: $\int x^m P_q[x^n] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $m-n+1=0$

Derivation: Integration by substitution

Basis: $x^{n-1} F[x^n] = \frac{1}{n} \text{Subst}[F[x], x, x^n]\, \partial_x x^n$

Rule: If $m-n+1=0$, then

$$\int x^m P_q[x^n] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \to \frac{1}{n} \text{Subst}\left[\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx,\ x,\ x^n\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[SubstFor[x^n,Pq,x]*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && EqQ[Simplify[m-n+1],0]
```

2: $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \to \int \text{ExpandIntegrand}\left[(d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p,\ x\right] dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && IGtQ[p,0]
```

3: $\int (g\,x)^m \left(d+e\,x^n+f\,x^{2\,n}\right) \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $a\,e\,(m+1)-b\,d\,(m+n\,(p+1)+1)=0\,\land\,a\,f\,(m+1)-c\,d\,(m+2\,n\,(p+1)+1)=0\,\land\,m\neq-1$

- Rule: If $a\,e\,(m+1)-b\,d\,(m+n\,(p+1)+1)=0\,\land\,a\,f\,(m+1)-c\,d\,(m+2\,n\,(p+1)+1)=0\,\land\,m\neq-1$, then

$$\int (g\,x)^m \left(d+e\,x^n+f\,x^{2\,n}\right) \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \longrightarrow \frac{d\,(g\,x)^{m+1} \left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{a\,g\,(m+1)}$$

- Program code:

```
Int[(g_.*x_)^m_.*(d_+e_.*x_^n_.+f_.*x_^n2_.)*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*(g*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*g*(m+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[n2,2*n] && EqQ[a*e*(m+1)-b*d*(m+n*(p+1)+1),0] && EqQ[a*f*(m+1)-c*d*(m+2*n*(p+1)+1),0] && NeQ[m,-1]
```

```
Int[(g_.*x_)^m_.*(d_+f_.*x_^n2_.)*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*(g*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*g*(m+1)) /;
FreeQ[{a,b,c,d,f,g,m,n,p},x] && EqQ[n2,2*n] && EqQ[m+n*(p+1)+1,0] && EqQ[c*d+a*f,0] && NeQ[m,-1]
```

4: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\;\land\;p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$, then $\partial_x\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(b+2\,c\,x^n\right)^{2\,p}}=0$

Basis: If $b^2-4\,a\,c=0$, then $\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(b+2\,c\,x^n\right)^{2\,p}}=\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,\left(b+2\,c\,x^n\right)^{2\,\text{FracPart}[p]}}$

Rule: If $b^2-4\,a\,c=0\;\land\;p\notin\mathbb{Z}$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,\left(b+2\,c\,x^n\right)^{2\,\text{FracPart}[p]}}\int (d\,x)^m\,P_q[x]\,\left(b+2\,c\,x^n\right)^{2\,p}\,dx$$

–

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^n)^(2*FracPart[p]))*Int[(d*x)^m*Pq*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

5. $\int (d\,x)^m P_q[x^n]\,(a+b\,x^n+c\,x^{2n})^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m P_q[x^n]\,(a+b\,x^n+c\,x^{2n})^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1} F[x],\,x,\,x^n\right]\partial_x x^n$

Note: If $n\in\mathbb{Z}\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$, then $m\in\mathbb{Z}$, and $(d\,x)^m$ automatically evaluates to $d^m\,x^m$.

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m P_q[x^n]\,(a+b\,x^n+c\,x^{2n})^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1} P_q[x]\,(a+b\,x+c\,x^2)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*SubstFor[x^n,Pq,x]*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(d\,x)^m}{x^m}=0$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\to\;\frac{(d\,x)^m}{x^m}\int x^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(d_*x_)^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

6: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $P_q[x,\,0]=0$

Derivation: Algebraic simplification

Rule: If $P_q[x,\,0]=0$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\to\;\frac{1}{d}\int (d\,x)^{m+1}\,\text{PolynomialQuotient}[P_q[x],\,x,\,x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/d*Int[(d*x)^(m+1)*PolynomialQuotient[Pq,x,x]*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[Coeff[Pq,x,0],0]
```

7. $\int \frac{(d\,x)^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx$ when $b^2 - 4\,a\,c == 0 \land 2\,m - n + 2 == 0 \land c\,e + a\,h == 0$

1: $\int \frac{x^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx$ when $b^2 - 4\,a\,c == 0 \land 2\,m - n + 2 == 0 \land c\,e + a\,h == 0$

Rule: If $b^2 - 4\,a\,c == 0 \land 2\,m - n + 2 == 0 \land c\,e + a\,h == 0$, then

$$\int \frac{x^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx \longrightarrow -\frac{2\,c\,(b\,f - 2\,a\,g) + 2\,h\,(b^2 - 4\,a\,c)\,x^{n/2} + 2\,c\,(2\,c\,f - b\,g)\,x^n}{c\,n\,(b^2 - 4\,a\,c)\,\sqrt{a + b\,x^n + c\,x^{2\,n}}}$$

Program code:

```
Int[x_^m_.*(e_+f_.*x_^q_.+g_.*x_^r_.+h_.*x_^s_.)/(a_+b_.*x_^n_.+c_.*x_^n2_.)^(3/2),x_Symbol] :=
  -(2*c*(b*f-2*a*g)+2*h*(b^2-4*a*c)*x^(n/2)+2*c*(2*c*f-b*g)*x^n)/(c*n*(b^2-4*a*c)*Sqrt[a+b*x^n+c*x^(2*n)]) /;
FreeQ[{a,b,c,e,f,g,h,m,n},x] && EqQ[n2,2*n] && EqQ[q,n/2] && EqQ[r,3*n/2] && EqQ[s,2*n] &&
  NeQ[b^2-4*a*c,0] && EqQ[2*m-n+2,0] && EqQ[c*e+a*h,0]
```

2: $\int \frac{(d\,x)^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx$ when $b^2 - 4\,a\,c == 0 \land 2\,m - n + 2 == 0 \land c\,e + a\,h == 0$

Rule: If $b^2 - 4\,a\,c == 0 \land 2\,m - n + 2 == 0 \land c\,e + a\,h == 0$, then

$$\int \frac{(d\,x)^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx \longrightarrow \frac{(d\,x)^m}{x^m}\int \frac{x^m \left(e + f\,x^{n/2} + g\,x^{3\,n/2} + h\,x^{2\,n}\right)}{\left(a + b\,x^n + c\,x^{2\,n}\right)^{3/2}}\,dx$$

Program code:

```
Int[(d_*x_)^m_.*(e_+f_.*x_^q_.+g_.*x_^r_.+h_.*x_^s_.)/(a_+b_.*x_^n_.+c_.*x_^n2_.)^(3/2),x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*(e+f*x^(n/2)+g*x^((3*n)/2)+h*x^(2*n))/(a+b*x^n+c*x^(2*n))^(3/2),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && EqQ[n2,2*n] && EqQ[q,n/2] && EqQ[r,3*n/2] && EqQ[s,2*n] &&
  NeQ[b^2-4*a*c,0] && EqQ[2*m-n+2,0] && EqQ[c*e+a*h,0]
```

8. $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}$

1. $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$

1: $\int x^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m\in\mathbb{Z}^-$

Derivation: Algebraic expansion and trinomial recurrence 2b applied $n-1$ times

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,m\in\mathbb{Z}^-$, let $Q_{q-2\,n}[x]=\text{PolynomialQuotient}\left[x^m P_q[x],\,a+b\,x^n+c\,x^{2\,n},\,x\right]$ and $R_{2\,n-1}[x]=\text{PolynomialRemainder}\left[x^m P_q[x],\,a+b\,x^n+c\,x^{2\,n},\,x\right]$, then

$$\int x^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx\,\longrightarrow$$

$$\int R_{2\,n-1}[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx+\int Q_{q-2\,n}[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1} dx\,\longrightarrow$$

$$\begin{aligned}
&-\left(\left(x \left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\sum_{i=0}^{n-1}\left(\left(\left(b^2-2\,a\,c\right) R_{2\,n-1}[x,\,i]-a\,b\,R_{2\,n-1}[x,\,n+i]\right) x^i+c\left(b\,R_{2\,n-1}[x,\,i]-2\,a\,R_{2\,n-1}[x,\,n+i]\right) x^{n+i}\right)\right)\Big/\left(a\,n\,(p+1)\left(b^2-4\,a\,c\right)\right)\right)+\\
&\frac{1}{a\,n\,(p+1)\left(b^2-4\,a\,c\right)}\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\Bigg(a\,n\,(p+1)\left(b^2-4\,a\,c\right) x^{-m}\,Q_{q-2\,n}[x]+\\
&\sum_{i=0}^{n-1}\left(\left(\left(b^2\left(n\,(p+1)+i+1\right)-2\,a\,c\left(2\,n\,(p+1)+i+1\right)\right) R_{2\,n-1}[x,\,i]-a\,b\,(i+1)\,R_{2\,n-1}[x,\,n+i]\right) x^{i-m}+\right.\\
&\left.c\left(n\,(2\,p+3)+i+1\right)\left(b\,R_{2\,n-1}[x,\,i]-2\,a\,R_{2\,n-1}[x,\,n+i]\right) x^{n+i-m}\right)\Bigg)\,dx
\end{aligned}$$

–

Program code:

```
Int[x_^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x]},
  Module[{Q=PolynomialQuotient[a*(b*c)^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n+c*x^(2*n),x],
          R=PolynomialRemainder[a*(b*c)^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n+c*x^(2*n),x],i},
  -x*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*
    Sum[((b^2-2*a*c)*Coeff[R,x,i]-a*b*Coeff[R,x,n+i])*x^i+
      c*(b*Coeff[R,x,i]-2*a*Coeff[R,x,n+i])*x^(n+i),{i,0,n-1}] +
   1/(a*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*Int[x^m*(a+b*x^n+c*x^(2*n))^(p+1)*
    ExpandToSum[n*(p+1)*(b^2-4*a*c)*x^(-m)*Q+
      Sum[((b^2*(n*(p+1)+i+1)/a-2*c*(2*n*(p+1)+i+1))*Coeff[R,x,i]-b*(i+1)*Coeff[R,x,n+i])*x^(i-m)+
        c*(n*(2*p+3)+i+1)*(b/a*Coeff[R,x,i]-2*Coeff[R,x,n+i])*x^(n+i-m),{i,0,n-1}],x],x]] /;
 GeQ[q,2*n]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] && ILtQ[m,0]
```

2. $\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$

1: $\int x^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,m\in\mathbb{Z}\,\wedge\,\text{GCD}[m+1,\,n]\neq 1$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$, let $g=\text{GCD}[m+1,\,n]$, then $x^m\,F[x^n]=\frac{1}{g}\,\text{Subst}\left[x^{\frac{m+1}{g}-1}\,F\left[x^{\frac{n}{g}}\right],\,x,\,x^g\right]\,\partial_x x^g$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,m\in\mathbb{Z}$, let $g=\text{GCD}[m+1,\,n]$, if $g\neq 1$, then

$$\int x^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\rightarrow\,\frac{1}{g}\,\text{Subst}\left[\int x^{\frac{m+1}{g}-1}\,P_q\left[x^{\frac{n}{g}}\right]\,\left(a+b\,x^{\frac{n}{g}}+c\,x^{\frac{2\,n}{g}}\right)^p\,dx,\,x,\,x^g\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{g=GCD[m+1,n]},
  1/g*Subst[Int[x^((m+1)/g-1)*ReplaceAll[Pq,x->x^(1/g)]*(a+b*x^(n/g)+c*x^(2*n/g))^p,x],x,x^g] /;
 NeQ[g,1]] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[m]
```

2: $\int \frac{(d\,x)^m\,P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$

Derivation: Algebraic expansion

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,\text{NiceSqrtQ}\left[b^2-4\,a\,c\right]$, then

$$\int \frac{(d\,x)^m\,P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\int \text{ExpandIntegrand}\left[\frac{(d\,x)^m\,P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && NiceSqrtQ[b^2-4*a*c]
```

3: $\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\geq 2\,n\,\wedge\,m+q+2\,n\,p+1\neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A=0$, $B=1$ and $m=m-n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,q\geq 2\,n\,\wedge\,m+q+2\,n\,p+1\neq 0$, then

$$\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\int (d\,x)^m\,\left(P_q[x^n]-P_q[x,\,q]\,x^q\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx+\frac{P_q[x,\,q]}{d^q}\int (d\,x)^{m+q}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,\,q]\;(d\,x)^{m+q-2\,n+1}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{c\,d^{q-2\,n+1}\,(m+q+2\,n\,p+1)}+$$
$$\int (d\,x)^m\left(P_q\left[x^n\right]-P_q[x,\,q]\;x^q-\frac{P_q[x,\,q]\;\left(a\,(m+q-2\,n+1)\;x^{q-2\,n}+b\,(m+q+n\,(p-1)+1)\;x^{q-n}\right)}{c\,(m+q+2\,n\,p+1)}\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    Pqq*(d*x)^(m+q-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*d^(q-2*n+1)*(m+q+2*n*p+1)) +
    Int[(d*x)^m*ExpandToSum[Pq-Pqq*x^q-Pqq*(a*(m+q-2*n+1)*x^(q-2*n)+b*(m+q+n*(p-1)+1)*x^(q-n))/(c*(m+q+2*n*p+1)),x]*
      (a+b*x^n+c*x^(2*n))^p,x]] /;
  GeQ[q,2*n] && NeQ[m+q+2*n*p+1,0] && (IntegerQ[2*p] || EqQ[n,1] && IntegerQ[4*p] || IntegerQ[p+(q+1)/(2*n)])] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

3: $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,\neg\,\text{PolynomialQ}\left[P_q[x],\,x^n\right]$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}^+$, then $P_q[x] = \sum_{j=0}^{n-1} x^j \sum_{k=0}^{(q-j)/n+1} P_q[x,\,j+k\,n]\,x^{k\,n}$

Note: This rule transform integrand into a sum of terms of the form $(d\,x)^k Q_r[x^n] \left(a+b\,x^n+c\,x^{2\,n}\right)^p$.

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,\neg\,\text{PolynomialQ}\left[P_q[x],\,x^n\right]$, then

$$\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \,\longrightarrow\, \int \sum_{j=0}^{n-1} \frac{1}{d^j} (d\,x)^{m+j} \left(\sum_{k=0}^{(q-j)/n+1} P_q[x,\,j+k\,n]\,x^{k\,n}\right) \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],j,k},
  Int[Sum[1/d^j*(d*x)^(m+j)*Sum[Coeff[Pq,x,j+k*n]*x^(k*n),{k,0,(q-j)/n+1}]*(a+b*x^n+c*x^(2*n))^p,{j,0,n-1}],x]] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[PolyQ[Pq,x^n]]
```

4: $\int \frac{(d\,x)^m\,P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+$, then

$$\int \frac{(d\,x)^m\,P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\int \text{RationalFunctionExpand}\left[\frac{(d\,x)^m\,P_q[x]}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  Int[RationalFunctionExpand[(d*x)^m*Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

2. $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0 \land n\in\mathbb{Z}^-$

1. $\int (d\,x)^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0 \land n\in\mathbb{Z}^- \land m\in\mathbb{Q}$

1: $\int x^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0 \land n\in\mathbb{Z}^- \land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $F[x] == -\text{Subst}\left[\frac{F[x^{-1}]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Note: $x^q P_q[x^{-1}]$ is a polynomial in x.

Rule: If $b^2-4\,a\,c\neq 0 \land n\in\mathbb{Z}^- \land m\in\mathbb{Z}$, then

$$\int x^m P_q[x] \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx \longrightarrow -\text{Subst}\left[\int \frac{x^q P_q[x^{-1}] \left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p}{x^{m+q+2}} \, dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  -Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x→x^(-1)],x]*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+q+2),x],x,1/x]] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g>1$, then $(d\,x)^m\,F[x]=-\frac{g}{d}\,\text{Subst}\left[\frac{F\left[d^{-1}\,x^{-g}\right]}{x^{g\,(m+1)+1}},\,x,\,\frac{1}{(d\,x)^{1/g}}\right]\,\partial_x\,\frac{1}{(d\,x)^{1/g}}$

Note: $x^{g\,q}\,P_q\left[d^{-1}\,x^{-g}\right]$ is a polynomial in x.

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\in\mathbb{F}$, let $g=\text{Denominator}[m]$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,-\frac{g}{d}\,\text{Subst}\left[\int\frac{x^{g\,q}\,P_q\left[d^{-1}\,x^{-g}\right]\,\left(a+b\,d^{-n}\,x^{-g\,n}+c\,d^{-2\,n}\,x^{-2\,g\,n}\right)^p}{x^{g\,(m+q+1)+1}}\,dx,\,x,\,\frac{1}{(d\,x)^{1/g}}\right]$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[m],q=Expon[Pq,x]},
  -g/d*Subst[Int[ExpandToSum[x^(g*q)*ReplaceAll[Pq,x→d^(-1)*x^(-g)],x]*
    (a+b*d^(-n)*x^(-g*n)+c*d^(-2*n)*x^(-2*g*n))^p/x^(g*(m+q+1)+1),x],x,1/(d*x)^(1/g)]] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left(\frac{(d\,x)^m}{\left(x^{-1}\right)^m}\right)=0$

Basis: $F[x]=-\text{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Note: $x^q\,P_q\left[x^{-1}\right]$ is a polynomial in x.

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\,\wedge\,m\notin\mathbb{Q}$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,(d\,x)^m\,\left(x^{-1}\right)^m\int\frac{P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(x^{-1}\right)^m}\,dx$$

$$\longrightarrow\,-(d\,x)^m\,\left(x^{-1}\right)^m\,\text{Subst}\left[\int\frac{x^q\,P_q\left[x^{-1}\right]\,\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p}{x^{m+q+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(d_.*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  -(d*x)^m*(x^(-1))^m*Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x→x^(-1)],x]*(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+q+2),x],x,1/x]] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

9. $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$

1: $\int x^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $x^m\,P_q[x]\,F[x^n] = g\,\text{Subst}\left[x^{g\,(m+1)-1}\,P_q[x^g]\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\,\partial_x\,x^{1/g}$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$, let $g=\text{Denominator}[n]$, then

$$\int x^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;g\,\text{Subst}\left[\int x^{g\,(m+1)-1}\,P_q\left[x^g\right]\,\left(a+b\,x^{g\,n}+c\,x^{2\,g\,n}\right)^p\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g*(m+1)-1)*ReplaceAll[Pq,x->x^g]*(a+b*x^(g*n)+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

2. $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}$

1. $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}\,\wedge\,m-\frac{1}{2}\in\mathbb{Z}$

1: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}\,\wedge\,m+\frac{1}{2}\in\mathbb{Z}^+$

Derivation: Piecewise constant extraction

Basis: $\partial_x\,\frac{\sqrt{d\,x}}{\sqrt{x}} = 0$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}\,\wedge\,m+\frac{1}{2}\in\mathbb{Z}^+$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\to\;\frac{d^{m-\frac{1}{2}}\,\sqrt{d\,x}}{\sqrt{x}}\int x^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

–

Program code:

```
Int[(d_*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^(m-1/2)*Sqrt[d*x]/Sqrt[x]*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n] && IGtQ[m+1/2,0]
```

2: $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{F}\,\land\,m-\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\sqrt{x}}{\sqrt{d\,x}}==0$

–

Rule: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{F}\,\land\,m-\frac{1}{2}\in\mathbb{Z}^-$, then

$$\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\to\;\frac{d^{m+\frac{1}{2}}\,\sqrt{x}}{\sqrt{d\,x}}\int x^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

–

Program code:

```
Int[(d_*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^(m+1/2)*Sqrt[x]/Sqrt[d*x]*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n] && ILtQ[m-1/2,0]
```

2: $\int (d x)^m P_q[x] \left(a + b x^n + c x^{2 n}\right)^p dx$ when $b^2 - 4 a c \neq 0 \land n \in \mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(d x)^m}{x^m} = 0$

Rule: If $b^2 - 4 a c \neq 0 \land n \in \mathbb{F}$, then

$$\int (d x)^m P_q[x] \left(a + b x^n + c x^{2 n}\right)^p dx \longrightarrow \frac{(d x)^m}{x^m} \int x^m P_q[x] \left(a + b x^n + c x^{2 n}\right)^p dx$$

Program code:

```
Int[(d_*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

10. $\int (d\,x)^m\, P_q[x^n]\, \left(a + b\, x^n + c\, x^{2\,n}\right)^p\, dx$ when $b^2 - 4\,a\,c \neq 0\ \land\ \frac{n}{m+1} \in \mathbb{Z}$

1: $\int x^m\, P_q[x^n]\, \left(a + b\, x^n + c\, x^{2\,n}\right)^p\, dx$ when $b^2 - 4\,a\,c \neq 0\ \land\ \frac{n}{m+1} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1} \in \mathbb{Z}$, then $x^m\, F[x^n] = \frac{1}{m+1}\, \text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\, x,\, x^{m+1}\right] \partial_x x^{m+1}$

Rule: If $b^2 - 4\,a\,c \neq 0\ \land\ \frac{n}{m+1} \in \mathbb{Z}$

$$\int x^m\, P_q[x^n]\, \left(a + b\, x^n + c\, x^{2\,n}\right)^p\, dx \longrightarrow \frac{1}{m+1}\, \text{Subst}\left[\int P_q\left[x^{\frac{n}{m+1}}\right] \left(a + b\, x^{\frac{n}{m+1}} + c\, x^{\frac{2\,n}{m+1}}\right)^p dx,\, x,\, x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[ReplaceAll[SubstFor[x^n,Pq,x],x→x^Simplify[n/(m+1)]]*(a+b*x^Simplify[n/(m+1)]+c*x^Simplify[2*n/(m+1)])^p,x],x,x^(m+1
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(d\,x)^m}{x^m}==0$

Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (d\,x)^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{(d\,x)^m}{x^m}\int x^m\,P_q[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(d_*x_)^m_*Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

11. $\int (d\,x)^m\,P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p\in\mathbb{Z}^-$

1: $\int \frac{(d\,x)^m\,P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{q}\,\frac{1}{b-q+2\,c\,z}-\frac{2\,c}{q}\,\frac{1}{b+q+2\,c\,z}$

Rule: If $b^2-4\,a\,c\neq 0$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{(d\,x)^m\,P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\frac{2\,c}{q}\int \frac{(d\,x)^m\,P_q[x]}{b-q+2\,c\,x^n}\,dx-\frac{2\,c}{q}\int \frac{(d\,x)^m\,P_q[x]}{b+q+2\,c\,x^n}\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[(d*x)^m*Pq/(b-q+2*c*x^n),x] -
  2*c/q*Int[(d*x)^m*Pq/(b+q+2*c*x^n),x]] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0]
```

2: $\int (d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p \in \mathbb{Z}^-$

- Derivation: Algebraic expansion

- Rule: If $p \in \mathbb{Z}^-$, then

$$\int (d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p,\,x\right]dx$$

- Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && ILtQ[p+1,0]
```

X: $\int (d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$

- Rule:

$$\int (d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \int (d\,x)^m P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

- Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Unintegrable[(d*x)^m*Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && (PolyQ[Pq,x] || PolyQ[Pq,x^n])
```

S: $\int u^m P_q[v^n] \left(a + b v^n + c v^{2n}\right)^p dx$ when $v == f + g x \land u == h v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == h v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule: If $v == f + g x \land u == h v$, then

$$\int u^m P_q[v^n] \left(a + b v^n + c v^{2n}\right)^p dx \longrightarrow \frac{u^m}{g v^m} \text{Subst}\left[\int x^m P_q[x^n] \left(a + b x^n + c x^{2n}\right)^p dx, x, v\right]$$

Program code:

```
Int[u_^m_.*Pq_*(a_+b_.*v_^n_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*SubstFor[v,Pq,x]*(a+b*x^n+c*x^(2*n))^p,x],x,v] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && LinearPairQ[u,v,x] && PolyQ[Pq,v^n]
```